

```

clear
tempname pf
postfile `pf' mX1 sX1 skX1 kX1 mX2 sX2 skX2 kX2 using temp.dta, replace
forvalues iter=1/10000 {
drop _all
qui set obs 20
*-----
* Algorithm 1: Draw p then X
* Prior Dir(25,10,5) = Dir(40;0.625,0.250,0.125)
* Sample 20 X's
*-----
local g0 = rgamma(25,1)
local g1 = rgamma(10,1)
local g2 = rgamma(5,1)
local p0 = `g0'/(`g0'+`g1'+`g2')
local p1 = `g1'/(`g0'+`g1'+`g2')
local p2 = `g2'/(`g0'+`g1'+`g2')
gen u = runiform()
gen X = (u>`p0')+(u>(`p0'+`p1'))
qui su X ,detail
local mX1 = r(mean)
local sX1 = r(sd)
local skX1 = r(skewness)
local kX1 = r(kurtosis)

*-----
* Algorithm 2: Chinese restaurant
*-----
qui replace X = .
local u = runiform()
forvalues i=1/20 {
    local test = 40/(40+`i'-1)
    if runiform() < `test' {
        local u = runiform()
        qui replace X = (`u'>0.625)+(`u'>0.875) in `i'
    }
    else {
        local j = 1 + int(runiform()*(`i'-1))
        qui replace X = X[`j'] in `i'
    }
}
qui su X, detail
local mX2 = r(mean)
local sX2 = r(sd)
local skX2 = r(skewness)
local kX2 = r(kurtosis)
post `pf' (`mX1') (`sX1') (`skX1') (`kX1') (`mX2') (`sX2') (`skX2')
(`kX2')
}
postclose `pf'
*-----
* Missing values correspond to random samples
* consisting of 20 identical values
*-----
use temp.dta, clear
su *

```